clear all
set more off
set maxvar 32000
set matsize 11000
*************************************************************************************************
*set scheme aspen, perm
grstyle init
grstyle set plain, horizontal grid

use "$folder/data/data_ready.dta", clear
cd "$folder/output/tables"
local x=18

xtset state year, yearly delta(2)

label variable growth_gsp "Growth GSP"
label variable log_real1gsppc "Log Real Income per capita"
label variable growth_pc "Growth Real Income per capita"
label variable lestab_D "Growth Number Estabilishments"
label variable lprofits_mill_D "Growth State Profits, Millions"
label variable lcbp_emp_D "Growth Employment"

label variable log_provisions_18 "Legislative Detail"
label variable log_cond_list_prov_18 "Contingent Provisions" 
label variable log_non_cond_list_prov_18 "Non-Contingent Provisions" 
label variable shock_prov18 "Shock to Legislative Detail"
label variable shock_cond_list_prov18 "Shock to Contingencies"
label variable shock_non_cond_list_prov18 "Shock to Non-Contingencies"

label variable gsp_initial "Initial Gross State Product"
label variable growth_initial "Initial GSP Growth"
label variable gsp_pc_initial "Initial GSP per capita"
label variable lestab_D "Growth Number Establishments"
label variable lprofits_mill_D "Growth State Profits, Millions"
label variable lcbp_emp_D "Growth Employment"

*******************
* MAIN 2SLS TABLE
*******************

* 1 // year FE, state FE
ivreghdfe growth_pc (log_provisions_`x' = shock_prov`x') , absorb(state year) cl(state)
outreg2 using Table_3_2SLS, excel replace label  ctitle(2SLS)  nocons addtext(State FE, X, Time FE, X) addstat(First Stage F-stat, e(rkf), R-squared,  e(r2)) alpha(0.01, 0.05, 0.1) symbol(**, *, +) nor2 nonotes addnote(SE clustered by `e(clustvar)', **p<.01; *p<.05; +p<.1)

* 2 // year, state, state trend
ivreghdfe growth_pc  (log_provisions_`x' = shock_prov`x'), absorb(state##c.year year)  cl(state)
outreg2 using Table_3_2SLS, excel append label  ctitle(2SLS)  nocons addtext(State FE, X, Time FE, X, State-Specific Trends, X) addstat(First Stage F-stat, e(rkf), R-squared,  e(r2)) alpha(0.01, 0.05, 0.1) symbol(**, *, +) nor2 nonotes addnote(SE clustered by `e(clustvar)', **p<.01; *p<.05; +p<.1)

* 3 // initial growth, initial GSP, and initial GSP per capita, interacted with year
ivreghdfe growth_pc  (log_provisions_`x' = shock_prov`x'), absorb(state year##c.(gsp_initial  growth_initial gsp_pc_initial)) cl(state)
outreg2 using Table_3_2SLS, excel append label  ctitle(2SLS)  nocons addtext(State FE, X, Time FE, X, Initial Econ x Time, X) addstat(First Stage F-stat, e(rkf), R-squared,  e(r2)) alpha(0.01, 0.05, 0.1) symbol(**, *, +) nor2 nonotes addnote(SE clustered by `e(clustvar)', **p<.01; *p<.05; +p<.1)

* 4 // initial sector shares (4 PCs), interacted with year, p=.04
ivreghdfe growth_pc  (log_provisions_`x' = shock_prov`x'), absorb(state  year##c.(pc1_sectors pc2_sectors pc3_sectors pc4_sectors)) cl(state)
outreg2 using Table_3_2SLS, excel append label ctitle(2SLS)  nocons addtext(State FE, X, Time FE, X, PC Sector Shares x Time, X) addstat(First Stage F-stat, e(rkf), R-squared,  e(r2)) alpha(0.01, 0.05, 0.1) symbol(**, *, +) nor2 nonotes addnote(SE clustered by `e(clustvar)', **p<.01; *p<.05; +p<.1)

* 5 // initial population / demographics X year
ivreghdfe growth_pc  (log_provisions_`x' = shock_prov`x') , absorb(state year##c.(initial_lpop initial_lpop2  initial_urban initial_foreign)) cl(state)
outreg2 using Table_3_2SLS, excel append label ctitle(2SLS)  nocons addtext(State FE, X, Time FE, X, Pre-Period Demog x Time, X) addstat(First Stage F-stat, e(rkf), R-squared,  e(r2)) alpha(0.01, 0.05, 0.1) symbol(**, *, +) nor2 nonotes addnote(SE clustered by `e(clustvar)', **p<.01; *p<.05; +p<.1)

* 6 // current-period topic controls
ivreghdfe growth_pc  (log_provisions_`x' = shock_prov`x'), absorb(state year one#c.(share`x'_provision*)) cl(state)
outreg2 using Table_3_2SLS, excel append label  ctitle(2SLS)  nocons addtext(State FE, X, Time FE, X, Topic Share, X) addstat(First Stage F-stat, e(rkf), R-squared,  e(r2)) alpha(0.01, 0.05, 0.1) symbol(**, *, +) nor2 nonotes addnote(SE clustered by `e(clustvar)', **p<.01; *p<.05; +p<.1)

* 7 // all
ivreghdfe growth_pc  (log_provisions_`x' = shock_prov`x'), absorb(state##c.year year##c.(gsp_initial  growth_initial gsp_pc_initial pc1_sectors pc2_sectors pc3_sectors pc4_sectors initial_lpop initial_lpop2  initial_urban initial_foreign) one#c.(share`x'_provision* L.logexp L.logexp2 L.loglegexp L.loglegexp2 L.growth_pc)) cl(state)
outreg2 using Table_3_2SLS, excel append label  ctitle(2SLS)  nocons addtext(State FE, X, Time FE, X, State-Specific Trends, X, Initial Econ x Time, X, PC Sector Shares x Time, X, Pre-Period Demog x Time, X, Govt Expenditure, X, Topic Share, X, Lagged Growth, X) addstat(First Stage F-stat, e(rkf), R-squared,  e(r2)) alpha(0.01, 0.05, 0.1) symbol(**, *, +) nor2 nonotes addnote(SE clustered by `e(clustvar)', **p<.01; *p<.05; +p<.1)
